

**CCOS****PURPOSE**

Compute the real or complex component of the cosine of a complex number.

**SYNTAX 1**

LET <yr> = CCOS(<xr>,<xc>) <SUBSET/EXCEPT/FOR qualification>

where <xr> is a number, parameter, or variable that specifies the real component of the the complex number;

<xc> is a number, parameter, or variable that specifies the complex component of the the complex number;

<yr> is a variable or a parameter (depending on what <xr> and <xc> are) where the real component of the computed cosine value is stored;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the real component of the complex cosine function.

**SYNTAX 2**

LET <yc> = CCOSI(<xr>,<xc>) <SUBSET/EXCEPT/FOR qualification>

where <xr> is a number, parameter, or variable that specifies the real component of the the complex number;

<xc> is a number, parameter, or variable that specifies the complex component of the the complex number;

<yc> is a variable or a parameter (depending on what <xr> and <xc> are) where the complex component of the computed cosine value is stored;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the complex component of the complex cosine function.

**EXAMPLES**

LET AR = CCOS(-2,1)

LET AC = CCOSI(-2,1)

LET ZR = CCOS(XR,XC)

LET ZC = CCOSI(XR,XC)

**NOTE**

DATAPLOT uses the Fortran intrinsic function CCOS to compute this function.

**DEFAULT**

None

**SYNONYMS**

None

**RELATED COMMANDS**

COS	=	Compute the cosine of a real number.
CABS	=	Compute the absolute value of a complex number.
CEXP	=	Compute the real component of the exponential of a complex number.
CEXPI	=	Compute the complex component of the exponential of a complex number.
CLOG	=	Compute the real component of the logarithm of a complex number.
CLOGI	=	Compute the complex component of the logarithm of a complex number.
CSIN	=	Compute the real component of the sine of a complex number.
CSINI	=	Compute the complex component of the sine of a complex number.
CSQRT	=	Compute the real component of the square root of a complex number.
CSQRTI	=	Compute the complex component of the square root of a complex number.

**APPLICATIONS**

Elementary functions

**IMPLEMENTATION DATE**

94/10

## PROGRAM

```

X1LABEL SOLID = REAL COMPONENT
X2LABEL DASH = COMPLEX COMPONENT
LINE SOLID DASH
YLIMITS -3 3
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
LET C = PI/4
TITLE CCOS, COMPLEX COMPONENT = ^C
PLOT CCOS(X,C) FOR X = -10 0.1 10 AND
PLOT CCOSI(X,C) FOR X = -10 0.1 10
LET C = -PI/4
TITLE CCOS, COMPLEX COMPONENT = ^C
PLOT CCOS(X,C) FOR X = -10 0.1 10 AND
PLOT CCOSI(X,C) FOR X = -10 0.1 10
LET C = PI/2
TITLE CCOS, COMPLEX COMPONENT = ^C
PLOT CCOS(X,C) FOR X = -10 0.1 10 AND
PLOT CCOSI(X,C) FOR X = -10 0.1 10
LET C = -PI/2
TITLE CCOS, COMPLEX COMPONENT = ^C
PLOT CCOS(X,C) FOR X = -10 0.1 10 AND
PLOT CCOSI(X,C) FOR X = -10 0.1 10
END OF MULTIPLY

```

